

NAG Library Routine Document

G01DCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01DCF computes an approximation to the variance-covariance matrix of an ordered set of independent observations from a Normal distribution with mean 0.0 and standard deviation 1.0.

2 Specification

```
SUBROUTINE G01DCF (N, EXP1, EXP2, SUMSSQ, VEC, IFAIL)
INTEGER N, IFAIL
REAL (KIND=nag_wp) EXP1, EXP2, SUMSSQ, VEC(N*(N+1)/2)
```

3 Description

G01DCF is an adaptation of the Applied Statistics Algorithm AS 128, see Davis and Stephens (1978). An approximation to the variance-covariance matrix, V , using a Taylor series expansion of the Normal distribution function is discussed in David and Johnson (1954).

However, convergence is slow for extreme variances and covariances. The present routine uses the David–Johnson approximation to provide an initial approximation and improves upon it by use of the following identities for the matrix.

For a sample of size n , let m_i be the expected value of the i th largest order statistic, then:

- (a) for any $i = 1, 2, \dots, n$, $\sum_{j=1}^n V_{ij} = 1$
- (b) $V_{12} = V_{11} + m_n^2 - m_n m_{n-1} - 1$
- (c) the trace of V is $tr(V) = n - \sum_{i=1}^n m_i^2$
- (d) $V_{ij} = V_{ji} = V_{rs} = V_{sr}$ where $r = n + 1 - i$, $s = n + 1 - j$ and $i, j = 1, 2, \dots, n$. Note that only the upper triangle of the matrix is calculated and returned column-wise in vector form.

4 References

David F N and Johnson N L (1954) Statistical treatment of censored data, Part 1. Fundamental formulae *Biometrika* **41** 228–240

Davis C S and Stephens M A (1978) Algorithm AS 128: approximating the covariance matrix of Normal order statistics *Appl. Statist.* **27** 206–212

5 Arguments

- | | |
|---|--------------|
| 1: N – INTEGER | <i>Input</i> |
| On entry: n , the sample size. | |
| Constraint: $N > 0$. | |
| 2: EXP1 – REAL (KIND=nag_wp) | <i>Input</i> |
| On entry: the expected value of the largest Normal order statistic, m_n , from a sample of size n . | |

- 3: EXP2 – REAL (KIND=nag_wp) *Input*
On entry: the expected value of the second largest Normal order statistic, m_{n-1} , from a sample of size n .
- 4: SUMSSQ – REAL (KIND=nag_wp) *Input*
On entry: the sum of squares of the expected values of the Normal order statistics from a sample of size n .
- 5: VEC($N \times (N + 1)/2$) – REAL (KIND=nag_wp) array *Output*
On exit: the upper triangle of the n by n variance-covariance matrix packed by column. Thus element V_{ij} is stored in $\text{VEC}(i + j \times (j - 1)/2)$, for $1 \leq i \leq j \leq n$.
- 6: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 1$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

For $n \leq 20$, where comparison with the exact values can be made, the maximum error is less than 0.0001.

8 Parallelism and Performance

G01DCF is not threaded in any implementation.

9 Further Comments

The time taken by G01DCF is approximately proportional to n^2 .

The arguments EXP1 ($= m_n$), EXP2 ($= m_{n-1}$) and SUMSSQ ($= \sum_{j=1}^n m_j^2$) may be found from the expected values of the Normal order statistics obtained from G01DAF (exact) or G01DBF (approximate).

10 Example

A program to compute the variance-covariance matrix for a sample of size 6. G01DAF is called to provide values for EXP1, EXP2 and SUMSSQ.

10.1 Program Text

```
Program g01dcfe
!
!     G01DCF Example Program Text
!
!     Mark 26 Release. NAG Copyright 2016.
!
!     .. Use Statements ..
Use nag_library, Only: g01daf, g01dcf, nag_wp
!
!     .. Implicit None Statement ..
Implicit None
!
!     .. Parameters ..
Integer, Parameter :: nout = 6
!
!     .. Local Scalars ..
Real (Kind=nag_wp) :: errest, etol, exp1, exp2, sumssq
Integer :: i, ifail, iw, j, k, lvec, n
!
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: pp(:), vec(:), work(:)
!
!     .. Executable Statements ..
Write (nout,*) 'G01DCF Example Program Results'
Write (nout,*)

!
!     Set the problem size
n = 6
etol = 0.0001E0_nag_wp

lvec = n*(n+1)/2
iw = 3*n/2
Allocate (pp(n),work(iw),vec(lvec))

!
!     Compute normal scores
ifail = 0
Call g01daf(n,pp,etol,errest,work,iw,ifail)

exp1 = pp(n)
exp2 = pp(n-1)
sumssq = 0.0E0_nag_wp
Do i = 1, n
    sumssq = sumssq + pp(i)*pp(i)
End Do

!
!     Compute approximate variance-covariance matrix
ifail = 0
Call g01dcf(n,exp1,exp2,sumssq,vec,ifail)

!
!     Display results
Write (nout,99999) 'Sample size = ', n
```

```
Write (nout,*) 'Variance-covariance matrix'
k = 1
Do j = 1, n
    Write (nout,99998) vec(k:(k+j-1))
    k = k + j
End Do

99999 Format (1X,A,I2)
99998 Format (1X,6F8.4)
End Program g01dcfe
```

10.2 Program Data

None.

10.3 Program Results

G01DCF Example Program Results

```
Sample size = 6
Variance-covariance matrix
  0.4159
  0.2085  0.2796
  0.1394  0.1889  0.2462
  0.1025  0.1397  0.1834  0.2462
  0.0774  0.1060  0.1397  0.1889  0.2796
  0.0563  0.0774  0.1025  0.1394  0.2085  0.4159
```
